
	Reason for change:
	1. In the cell FACH transistion testcases, a delay of 60ms (FDD) is provided for the UE to complete the FACH transistion. However, if the UE is able to complete the transition quicker and returns the RB Setup Complete whilst the SS is still in the midst of local end configurations the incoming message will be handled in default and result in FAIL verdict.

	Consequences if not approved:
	A Conformant UEs may fail Cell-FACH test cases

	
	

	Clauses affected:
	All Cell FACH transition testcases

	
	

	
	Y
	N
	
	

	Other specs
	
	x
	 Other core specifications

	

	affected:
	
	x
	 Test specifications
	

	(show related CRs)
	
	x
	 O&M Specifications
	

	
	

	Other comments:
	


Change 1
	Function name
	f_UTRAN_RB_SetUp_PS_RAB

	Reason for change
	In the cell FACH transistion testcases, a delay of 60ms (FDD) is provided for the UE to complete the FACH transistion. However, if the UE is able to complete the transition quicker and returns the RB Setup Complete whilst the SS is still in the midst of local end configurations the incoming message will be handled in default and result in FAIL verdict.

	Summary of change
	Called f_UTRAN_Default_BlockAM to be set to true before the local reconfiguration
f_UTRAN_Default_BlockAM to be set to FALSE after the local reconfiguration



	TTCN module
	UTRAN_RAB_Functions.ttcn

	MCC160 Comment
	


Before change
 function f_UTRAN_RB_SetUp_PS_RAB(UTRAN_CellId_Type p_CellId,

                                   RB_ConfigType     p_RbType,

                                   boolean           p_WaitForSetupCmpl := true,

                                   boolean           p_LoopbackRLC_PDU := false,  // @sic R5s130195 PDCP sic@

                                   boolean p_UseStoredNsapi := false)  //@sic R5s141127 sic@

    runs on UTRAN_PTC

  { /* @sic R5s120139 - Additional Changes sic@ */

    var UTRAN_FDD_TDD_Type v_FDD_TDD_Mode := f_UTRAN_CellInfo_GetFDD_TDD(p_CellId);

    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();

    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime(p_CellId);

    var ActivationTime v_Activationtime := v_CfnInfo.actTime; /* @sic R5s150684 sic@ */

    var template (value) SS_ActivationTime v_SS_ActivationTime := cs_ActivateCFN(v_Activationtime);

    var IntegrityCheckInfo v_Integrityinfo;

    var template (value) DL_DCCH_Message v_RB_SetUp;

    var C_RNTI v_New_CRNTI := '1010101010101010'B;

    var boolean v_CnfReq := true; //@sic R5s110210 Additional MCC160 changes sic@

    if (not p_WaitForSetupCmpl) {     // @sic R5s120139 sic@

      // estimate the start value since we cannot take it from the Setup Complete message

      v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);

      f_UTRAN_Security_Set(v_SecurityInfo);

    }

    v_Integrityinfo := v_SecurityInfo.dl_IntegrityCheckInfo; /* @sic R5s150684 sic@ */

    select (p_RbType) {

      case (cell_DCH_64kPS_RAB_SRB) {

        // Send RB_Set Up for DCH 64k PS for FDD

        v_RB_SetUp := f_Get_cs_108_RB_SetUp64k_PS(p_CellId, v_Integrityinfo, v_Activationtime, p_UseStoredNsapi);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp); //@sic R5s141127 sic@

        //Configure 64k PS RAB at SS

        f_SS_Config_64kPS_RAB (p_CellId, -, -, -, -, v_SS_ActivationTime, p_LoopbackRLC_PDU);  // @sic R5s130195 PDCP sic@

      }

      case (cell_DCH_HS_DSCH) {

        v_RB_SetUp := f_Get_cs_108_RB_SetUpHS_FromDCH_UL_64k(p_CellId, v_Integrityinfo, v_Activationtime);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp); //@sic R5s110210 Additional MCC160 changes sic@

        // Configure HSDPA (64k in UL) RAB at SS

        f_SS_Config_HSDPA_64kUL_RAB (p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU); // @sic R5s130195 PDCP sic@

      }

      case (cell_FACH_PS){

        // Send RB_Set Up

        v_RB_SetUp := f_Get_cs_108_RB_SetUpFACH_PS(v_Integrityinfo);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp, not v_CnfReq); //@sic R5s110210 Additional MCC160 changes sic@

        //Delay after sending the RAB SETUP in CELL FACH state; legacy of TTCN-2 suite; kept as exception only

        if (v_FDD_TDD_Mode == UTRAN_TDD) {

          f_Delay(tsc_DelaySetupFACH_TDD);  //@sic R5s140468 sic@

        } else {

          f_Delay(tsc_DelaySetupFACH);

        }

        //Reconfigure SS with the new C-RNTI

        f_UTRAN_SS_New_CRNTI(p_CellId, v_New_CRNTI);

        // Configure PDCP when there is no loopback of RLC PDU @sic R5s130195 PDCP sic@

        if (not(p_LoopbackRLC_PDU)) {

          f_SS_RB_ConfigurePDCP(p_CellId, tsc_RB20);

        }

        f_UTRAN_CellInfo_SetConfigType (p_CellId, cell_FACH_PS);

      }

      case (cell_DCH_E_HS) {

        FatalError(__FILE__, __LINE__, "Configuration not supported");

      }

      case else {

        FatalError(__FILE__, __LINE__, "Invalid configuration");

      }

    }

    if (p_WaitForSetupCmpl) {

      f_UTRAN_RRC_ReceiveRB_SetupCmpl_PS_RAB (p_CellId, p_RbType);

    }

    else {      // @sic R5s120139 sic@

      // just apply the ciphering at SS using the estimated start value

      f_UTRAN_SS_CipheringAM_RAB_UL_DL(p_CellId, ps_domain, p_RbType, v_SecurityInfo.authKeys.AuthPS.CK);

    }

  }
After change
  function f_UTRAN_RB_SetUp_PS_RAB(UTRAN_CellId_Type p_CellId,

                                   RB_ConfigType     p_RbType,

                                   boolean           p_WaitForSetupCmpl := true,

                                   boolean           p_LoopbackRLC_PDU := false,  // @sic R5s130195 PDCP sic@

                                   boolean p_UseStoredNsapi := false)  //@sic R5s141127 sic@

    runs on UTRAN_PTC

  { /* @sic R5s120139 - Additional Changes sic@ */

    var UTRAN_FDD_TDD_Type v_FDD_TDD_Mode := f_UTRAN_CellInfo_GetFDD_TDD(p_CellId);

    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();

    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime(p_CellId);

    var ActivationTime v_Activationtime := v_CfnInfo.actTime; /* @sic R5s150684 sic@ */

    var template (value) SS_ActivationTime v_SS_ActivationTime := cs_ActivateCFN(v_Activationtime);

    var IntegrityCheckInfo v_Integrityinfo;

    var template (value) DL_DCCH_Message v_RB_SetUp;

    var C_RNTI v_New_CRNTI := '1010101010101010'B;

    var boolean v_CnfReq := true; //@sic R5s110210 Additional MCC160 changes sic@

    if (not p_WaitForSetupCmpl) {     // @sic R5s120139 sic@

      // estimate the start value since we cannot take it from the Setup Complete message

      v_SecurityInfo.start_PS := f_CalculateStartValue_ConnectedMode(v_SecurityInfo.start_PS);

      f_UTRAN_Security_Set(v_SecurityInfo);

    }

    v_Integrityinfo := v_SecurityInfo.dl_IntegrityCheckInfo; /* @sic R5s150684 sic@ */

    select (p_RbType) {

      case (cell_DCH_64kPS_RAB_SRB) {

        // Send RB_Set Up for DCH 64k PS for FDD

        v_RB_SetUp := f_Get_cs_108_RB_SetUp64k_PS(p_CellId, v_Integrityinfo, v_Activationtime, p_UseStoredNsapi);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp); //@sic R5s141127 sic@

        //Configure 64k PS RAB at SS

        f_SS_Config_64kPS_RAB (p_CellId, -, -, -, -, v_SS_ActivationTime, p_LoopbackRLC_PDU);  // @sic R5s130195 PDCP sic@

      }

      case (cell_DCH_HS_DSCH) {

        v_RB_SetUp := f_Get_cs_108_RB_SetUpHS_FromDCH_UL_64k(p_CellId, v_Integrityinfo, v_Activationtime);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp); //@sic R5s110210 Additional MCC160 changes sic@

        // Configure HSDPA (64k in UL) RAB at SS

        f_SS_Config_HSDPA_64kUL_RAB (p_CellId, v_SS_ActivationTime, p_LoopbackRLC_PDU); // @sic R5s130195 PDCP sic@

      }

      case (cell_FACH_PS){

        // Send RB_Set Up

        v_RB_SetUp := f_Get_cs_108_RB_SetUpFACH_PS(v_Integrityinfo);

        f_UTRAN_RB_SetUp_Send(v_RB_SetUp, not v_CnfReq); //@sic R5s110210 Additional MCC160 changes sic@

        //Delay after sending the RAB SETUP in CELL FACH state; legacy of TTCN-2 suite; kept as exception only

        if (v_FDD_TDD_Mode == UTRAN_TDD) {

          f_Delay(tsc_DelaySetupFACH_TDD);  //@sic R5s140468 sic@

        } else {

          f_Delay(tsc_DelaySetupFACH);

        }

        //WA#R5s160xxx
        f_UTRAN_Default_BlockAM(true);
        //Reconfigure SS with the new C-RNTI

        f_UTRAN_SS_New_CRNTI(p_CellId, v_New_CRNTI);

        // Configure PDCP when there is no loopback of RLC PDU @sic R5s130195 PDCP sic@

        if (not(p_LoopbackRLC_PDU)) {

          f_SS_RB_ConfigurePDCP(p_CellId, tsc_RB20);

        }

        f_UTRAN_CellInfo_SetConfigType (p_CellId, cell_FACH_PS);

      }

      case (cell_DCH_E_HS) {

        FatalError(__FILE__, __LINE__, "Configuration not supported");

      }

      case else {

        FatalError(__FILE__, __LINE__, "Invalid configuration");

      }

    }

    if (p_WaitForSetupCmpl) {

      //WA#R5s161xxx

      f_UTRAN_Default_BlockAM(false); 

      f_UTRAN_RRC_ReceiveRB_SetupCmpl_PS_RAB (p_CellId, p_RbType);

    }

    else {      // @sic R5s120139 sic@

      // just apply the ciphering at SS using the estimated start value

      f_UTRAN_SS_CipheringAM_RAB_UL_DL(p_CellId, ps_domain, p_RbType, v_SecurityInfo.authKeys.AuthPS.CK);

      //WA#R5s161xxx

      f_UTRAN_Default_BlockAM(false);
    }

  }
Change 3
	Record name
	UTRAN_DefaultCtrl_Type

	Reason for change
	In the cell FACH transistion testcases, a delay of 60ms is provided for the UE to complete the FACH transistion. However, it has been observed that the UE is able to complete the transition quicker and returns the RB Setup Complete whilst the SS is still in the midst of local end configurations thereby causing the incoming message to be handled in default and result in FAIL verdict

	Summary of change
	Introduced a boolean flag.

	TTCN module
	UTRAN_Component.ttcn

	MCC160 Comment
	


Before change
 type record UTRAN_DefaultCtrl_Type {          /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_IRAT, POS, SSNITZ, UTRAN) */

    boolean U_TM_Blocked,

    boolean U_Dc_Blocked
  };

After change
 type record UTRAN_DefaultCtrl_Type {          /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_IRAT, POS, SSNITZ, UTRAN) */

    boolean U_TM_Blocked,

    boolean U_Dc_Blocked,

    boolean U_AM_Blocked    //WA#R5s160xxx
  };

Change 4
	Record name
	a_UTRAN_StandardDefault

	Reason for change
	In the cell FACH transistion testcases, a delay of 60ms is provided for the UE to complete the FACH transistion. However, it has been observed that the UE is able to complete the transition quicker and returns the RB Setup Complete whilst the SS is still in the midst of local end configurations thereby causing the incoming message to be handled in default and result in FAIL verdict

	Summary of change
	U_AM port message are blocked/allowed dpending on vc_UTRAN_Global.Ctrl.DefaultCtrl.U_AM_Blocked

Introduced a function for blocking incoming messages on U_AM port 

	TTCN module
	UTRAN_Component.ttcn

	MCC160 Comment
	


Before change
 altstep a_UTRAN_StandardDefault() runs on UTRAN_PTC

  { // @sic R5s110431 sic@

    // @sic R5s120049: new parameter to take TM out of the ports to be handled; this is e.g. to deal with an incoming RRC Connection Request directly after RRC Connection Release sic@

    // @sic R5s120086 (R5s120107): global variables instead of special default behaviours to support blocking of further ports (-> U_Dc) and avoid problems with the order of activating defaults sic@

    [] U_CPHY.receive (car_CPHY_Sync_IND (tsc_UL_DPCH1)) { // @sic R5s110216 sic@

      repeat;

    }

    [] U_CPHY.receive (car_CPHY_Out_of_Sync_IND (tsc_UL_DPCH1)) { // @sic R5s110216 sic@

      repeat;

    }

    [] U_CRLC.receive (car_CRLC_Integrity_Failure_IND) { // @sic R5s110216 sic@

      repeat;

    }

    [] U_AM.receive (cr_U_RLC_AM_IND_Any){

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_AM port");

    }

    [not vc_UTRAN_Global.Ctrl.DefaultCtrl.U_TM_Blocked] U_TM.receive (cr_RLC_TR_DATA_IND_Any) { // @sic R5s120098 R5s120086 sic@

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_TM port");

    }

    [] U_UM.receive (cr_RLC_UM_DATA_IND_Any){

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_UM port");

    }

    [] U_CPHY.receive (cr_U_CPHY_CONFIG_CNF_Any){

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CPHY port");

    }

    [] U_CMAC.receive (cr_U_CMAC_CONFIG_CNF_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CMAC port");

    }

    [] U_CRLC.receive (cr_U_CRLC_CONFIG_CNF_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CRLC port");

    }

    [not vc_UTRAN_Global.Ctrl.DefaultCtrl.U_Dc_Blocked] U_Dc.receive (cr_RRC_DATA_IND_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_Dc port");

    }

    // @sic R5s140316 sic@

    [] any timer.timeout {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected timeout");

    }

  }

After change
 altstep a_UTRAN_StandardDefault() runs on UTRAN_PTC

  { // @sic R5s110431 sic@

    // @sic R5s120049: new parameter to take TM out of the ports to be handled; this is e.g. to deal with an incoming RRC Connection Request directly after RRC Connection Release sic@

    // @sic R5s120086 (R5s120107): global variables instead of special default behaviours to support blocking of further ports (-> U_Dc) and avoid problems with the order of activating defaults sic@

    [] U_CPHY.receive (car_CPHY_Sync_IND (tsc_UL_DPCH1)) { // @sic R5s110216 sic@

      repeat;

    }

    [] U_CPHY.receive (car_CPHY_Out_of_Sync_IND (tsc_UL_DPCH1)) { // @sic R5s110216 sic@

      repeat;

    }

    [] U_CRLC.receive (car_CRLC_Integrity_Failure_IND) { // @sic R5s110216 sic@

      repeat;

    }

    [not vc_UTRAN_Global.Ctrl.DefaultCtrl.U_AM_Blocked] U_AM.receive (cr_U_RLC_AM_IND_Any){ //WA#R5s160xxx
      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_AM port");

    }

    [not vc_UTRAN_Global.Ctrl.DefaultCtrl.U_TM_Blocked] U_TM.receive (cr_RLC_TR_DATA_IND_Any) { // @sic R5s120098 R5s120086 sic@

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_TM port");

    }

    [] U_UM.receive (cr_RLC_UM_DATA_IND_Any){

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_UM port");

    }

    [] U_CPHY.receive (cr_U_CPHY_CONFIG_CNF_Any){

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CPHY port");

    }

    [] U_CMAC.receive (cr_U_CMAC_CONFIG_CNF_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CMAC port");

    }

    [] U_CRLC.receive (cr_U_CRLC_CONFIG_CNF_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_CRLC port");

    }

    [not vc_UTRAN_Global.Ctrl.DefaultCtrl.U_Dc_Blocked] U_Dc.receive (cr_RRC_DATA_IND_Any) {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected receive event at U_Dc port");

    }

    // @sic R5s140316 sic@

    [] any timer.timeout {

      all timer.stop;

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected timeout");

    }

  }  

New function:

  //WA#R5s160xxx
  function f_UTRAN_Default_BlockAM(boolean p_Block) runs on UTRAN_PTC

  {

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_AM_Blocked := p_Block;

  }

Change 5
	Record name
	f_UTRAN_ActivateDefault

	Reason for change
	In the cell FACH transistion testcases, a delay of 60ms is provided for the UE to complete the FACH transistion. However, it has been observed that the UE is able to complete the transition quicker and returns the RB Setup Complete whilst the SS is still in the midst of local end configurations thereby causing the incoming message to be handled in default and result in FAIL verdict

	Summary of change
	vc_UTRAN_Global.Ctrl.DefaultCtrl.U_AM_Blocked is initialised to false;

	TTCN module
	UTRAN_Component.ttcn

	MCC160 Comment
	


Before change
 function f_UTRAN_ActivateDefault(UTRAN_Default_Enum_Type p_Default) runs on UTRAN_PTC

  { // @sic R5s120049: new default behaviour "StandardDefault_BlockTM" (removed again due to R5s120086 (R5s120107)) sic@

    // @sic R5s120086 (R5s120107): global variables instead of special default behaviours to support blocking of further ports (-> U_Dc) and avoid problems with the order of activating defaults sic@

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_TM_Blocked := false;

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_Dc_Blocked := false;
    deactivate(vc_UTRAN_Global.Ctrl.Default);

    select (p_Default) {

      case (StandardDefault) {

        vc_UTRAN_Global.Ctrl.Default := activate(a_UTRAN_StandardDefault());

      }

      case else {

        FatalError (__FILE__, __LINE__, "unknown default behaviour");

      }

    }

  }
After change
 function f_UTRAN_ActivateDefault(UTRAN_Default_Enum_Type p_Default) runs on UTRAN_PTC

  { // @sic R5s120049: new default behaviour "StandardDefault_BlockTM" (removed again due to R5s120086 (R5s120107)) sic@

    // @sic R5s120086 (R5s120107): global variables instead of special default behaviours to support blocking of further ports (-> U_Dc) and avoid problems with the order of activating defaults sic@

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_TM_Blocked := false;

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_Dc_Blocked := false;

    vc_UTRAN_Global.Ctrl.DefaultCtrl.U_AM_Blocked := false; //WA#R5s160xxx
    deactivate(vc_UTRAN_Global.Ctrl.Default);

    select (p_Default) {

      case (StandardDefault) {

        vc_UTRAN_Global.Ctrl.Default := activate(a_UTRAN_StandardDefault());

      }

      case else {

        FatalError (__FILE__, __LINE__, "unknown default behaviour");

      }

    }

  }
